



## Ricominciamo da capo

MCmicrocomputer tratta l'argomento Computer Grafica sin dal primo numero, sia con una rubrica specifica, tenuta dal sottoscritto, sia con le prove delle apparecchiature specifiche per la Grafica, che sono sempre più numerose e a buon mercato, sia con le prove di apparecchiature non specifiche ma che fanno anche Computer Grafica, sia con le varie rubriche Software e Hardware dove spesso appaiono problematiche connesse con la Computer Grafica.

In effetti ormai tale argomento è talmente connesso con i micro e personal computer che risulta addirittura difficile isolarlo e considerarlo una materia a sé stante.

Per quanto riguarda la rubrica di Computer Grafica da me tenuta, nel corso dei vari articoli apparsi su MC non ho mai seguito un unico filo conduttore, ma ho presentato articoli su questo o quel problema affrontandolo e cercando di risolverlo nel giro di uno o due mesi, interessato, più che a fornire un programma da ricopiare e far girare, a stimolare un interesse che porti il lettore ad approfondire l'argomento per proprio conto ed usando le proprie macchine.

Con Marco Marinacci abbiamo discusso su come proseguire la rubrica, sia per tener conto dei nuovi lettori, parecchi dei quali sono probabilmente alle prime armi, sia per tener conto della attuale produzione di macchine, di software e della sua continua evoluzione.

Si è deciso, in un certo senso, di ricominciare da capo, facendo sulla Computer Grafica una trattazione continuativa, partendo dalle tematiche più semplici, che sono però propedeutiche rispetto alle successive, più complesse.

L'altra caratteristica che si darà a questa serie di articoli è che si cercherà di affrontare problematiche di C.G. indipendentemente dall'uso di questa o quella macchina, e in tale ottica i programmi a corredo degli articoli, esemplificativi come al solito dell'argomento, assumeranno il valore di mezzo con il quale esporre l'argomento e non quello di obiettivo finale dell'articolo.

Da un punto di vista mercato si possono notare due tendenze. Quella Apple che con Macintosh realizza una macchina in cui le funzioni grafiche e i suoi tipici strumenti (puntatore e videografico) sono integrati addirittura con il Sistema Operativo. Quello IBM che rende la grafica opzionale sul suo PC, ma accetta espansioni anche di grande potenza realizzate dai produttori indipendenti, e che sfruttano al meglio la potenza del 16 bit. La stessa IBM, tra l'altro, soprattutto con i nuovi annunci (terminale grafico 3270 PC, e PC grafico CG) apre nuove prospettive di utilizzo degli strumenti grafici soprattutto negli ambienti Centri di Elaborazione Dati, ove è già preminente la presenza della Number One.

Questo significa che in tali ambienti la Computer Grafica sarà produttiva in un sempre maggior numero di applicazioni e che le funzioni grafiche saranno affidate alla estesa capacità dei terminali e personal, collegati in rete tra di loro o con l'Host, per lo scambio dei dati.

La sensazione generale è comunque quella di una situazione in continua evoluzione tecnica ed economica.

L'applicazione e l'ambiente di lavoro

Occorre stabilire cosa si vuol fare (applicazione) e con quali mezzi si vuol farlo (l'ambiente di lavoro). Questa fase richiama tre modi di raggruppare temi di Computer Grafica:

### L'applicazione e l'ambiente di lavoro

— l'applicazione  
es. Game, Business Graphic, Animazione, ecc.

— gli strumenti hw, che si dividono in:  
strumenti input: tastiera, tavoletta, joystick, penna luminosa, mouse, ecc.

strumenti output: plotter, stampanti, videografico, ecc.

— gli strumenti software che a loro volta si dividono in:

- istruzioni di linguaggio,
- routine di linguaggio,
- tool,
- pacchetti grafici applicativi,
- pacchetti integrati.

Questa ultima suddivisione merita un chiarimento.

Il disegno di un cerchio sul video con talune macchine, che dispongono di linguaggi non estesi, non è possibile.

Con altre macchine è possibile utilizzando una istruzione propria del linguaggio. Con altre ancora è possibile eseguire il disegno, non tramite una specifica istruzione, ma con una routine che utilizza istruzioni differenti e che è più o meno complessa. Il tool è un software intermedio, che di per sé non risolve nessun problema ma serve come strumento per risolverlo. Un tipico tool è il generatore di caratteri che





Figure 1-2 - Programma Diecispirali - Output su video. Basta, all'interno di un loop sull'angolo, legare al variare di A, le altre grandezze, e si crea una figura anche complessa.

permette all'utente di costruirsi un proprio alfabeto per utilizzarlo quando e dove vuole.

Il pacchetto grafico applicativo è quello direttamente usabile senza dover intervenire a livello di programmazione. Un esempio è il software della tavoletta grafica di MC che permette di disegnare diretta-

mente sul video dell'Apple II. Il software integrato è quel software polifunzionale che permette, tra le altre cose, di fare grafica in maniera interattiva, direttamente dal suo ambiente operativo. Esempio è il Lotus 123, che può trarre direttamente da quelli presenti in sheet dati in forma Grafica.

Vi sono altri modi di raggruppare elementi di computer grafica, ad esempio:

- per tipo di problema
- grafica bidimensionale,
- grafica tridimensionale,
- grafica interattiva,
- ecc.

per tipo di prodotto finale

- figura su supporto cartaceo,
- figura su video,
- serie di figure su supporto video (animazione),
- serie di figure su video comandate da programma (grafica interattiva).

Risulta quindi difficile trattare la Computer Grafica suddividendola in argomenti logici separati, in quanto ognuno ne richiama altri. Per fare un esempio per realizzare un programma Flight Simulator, necessitiamo di certe apparecchiature hardware (videografico e joystick) e di certe funzioni software per generare sul video lo scenario (vista e cruscotto comandi) e trattarlo interattivamente in funzione della simulazione dei comandi ricevuti via joystick.

Analogamente libri di Game per Commodore 64, di Geometria per Apple II, di Business Graphics per IBM PC hanno sicuramente una buona parte di argomenti, quelli base, in comune.

## Il Sistema di Riferimento e gli Strumenti Software

Un altro argomento propedeutico rispetto a qualsiasi discorso applicativo di Computer Grafica è quello del sistema di riferimento entro il quale si lavora.

Risulta difficile fare anche un parallelo con il disegnatore il quale decide su cosa lavorare e con quali strumenti, in funzione del tipo di disegno che deve realizzare (tecnico, architettonico, artistico, ecc.).

In C.G. il sistema di riferimento e addirittura l'unità di misura da utilizzare appartengono a due mondi differenti a seconda che si usi il plotter per ottenere una uscita su carta sulla quale ha senso parlare di millimetri per le lunghezze o di gradi per gli angoli, o che si usi il video dove si usa il pixel e il sistema di riferimento lo riconosce come unità di misura così come riconosce come direzioni principali quelle del video.

Anche la funzione fondamentale (con la quale si generano tutte le altre) differisce profondamente. Sul video ha senso accendere o spegnere il singolo pixel che ha una sua consistenza fisica, al contrario non ha senso disegnare un punto con il plotter così come in geometria il punto è una entità

```

10 REM INIZIO
20 P = 3.1416:XC = 140:YC = 96:SY = .68:GOTO 100
30 REM ROUTINE FISSA
40 HGR2:HCOLOR=3:HPLOT 0,0 TO 279,0 TO 279,191 TO 0,191 TO 0,0
50 R = 0:SR = 1.2:RS = 4:SA = P / 12:AM = 20
60 HPLOT XC,YC:RETURN
70 GET SS:RETURN:REM RETURN PER CONTINUARE
100 REM SPIRALE UNO
110 GOSUB 30:AM = 15:FOR A = 0 TO AM STEP SA:R = R + SR
120 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
130 HPLOT TO XP,YP:NEXT A:GOSUB 70
200 REM SPIRALE DUE
210 GOSUB 30:FOR A = 0 TO AM STEP SA:R = P + SR
220 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
230 HPLOT TO XP,YP:HPLOT XC,YC TO XP,YP:NEXT A:GOSUB 70
300 REM SPIRALE TRE
310 GOSUB 30:FOR A = 0 TO AM STEP SA:R = R + SR
320 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
330 HPLOT TO XP,YP:HPLOT XC,YC TO XP,YP
340 XC = XC + RS * COS (A):YC = YC + RS * SIN (A)
350 NEXT A:GOSUB 70
400 REM SPIRALE QUATTRO
410 GOSUB 30:FOR A = 0 TO AM STEP SA:R = R + SR
420 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
430 HPLOT TO XP,YP:HPLOT XC,YC TO XP,YP
440 XC = XC + RS * COS (A + P):YC = YC + RS * SIN (A + P)
450 NEXT A:GOSUB 70
500 REM SPIRALE CINQUE
510 GOSUB 30:FOR A = 0 TO AM STEP SA:R = R + SR
520 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
530 HPLOT TO XP,YP:HPLOT XC,YC TO XP,YP
540 XC = XC + RS * COS (A):YC = YC + RS * SIN (A)
550 NEXT A:GOSUB 70
600 REM SPIRALE SEI
610 GOSUB 30:FOR A = 0 TO AM STEP SA:R = R + SR
620 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
630 XS = XC + R * COS (A + P):YS = YC + R * SIN (A + P) * SY
640 HPLOT XP,YP TO XS,YS:NEXT A:GOSUB 70
700 REM SPIRALE IPERBOLICA
710 GOSUB 30:CC = 50:AF = P / 8:AM = 4 * P
720 HPLOT XC,0 TO XC,191:HPLOT 0,YC TO 279,YC
730 FOR A = AM TO AF STEP -SA:R = CC / A
740 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
750 IF A = AM THEN HPLOT XP,YP
760 HPLOT TO XP,YP:NEXT A:GOSUB 70
800 REM SPIRALE SUPERIORE
810 GOSUB 30:XC = 40:YC = 40:AF = P / 6:M = 1.1:N = 1.2
820 HPLOT XC,0 TO XC,191:HPLOT 0,YC TO 279,YC
830 FOR A = AM TO AF STEP -SA:R = (CC / (A * M)) * N
840 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
850 IF A = AM THEN HPLOT XP,YP
860 HPLOT TO XP,YP:NEXT A:GOSUB 70
900 REM SPIRALE LOGARITMICA
910 GOSUB 30:XC = 120:YC = 96:AM = 4 * P:AF = P / 6:M = 6:N = 1.2
920 HPLOT XC,0 TO XC,191:HPLOT 0,YC TO 279,YC
930 FOR A = AM TO AF STEP -SA:R = M + A * N
940 XP = XC + R * COS (A):YP = YC + R * SIN (A) * SY
950 IF A = AM THEN HPLOT XP,YP
960 HPLOT TO XP,YP:NEXT A:GOSUB 70:TEXT:END

```

Figura 3 - Programma Diecispirali - Listato. Le varie spirali sono eseguite in sequenza. Per passare dall'una all'altra basta premere RETURN.

astratta senza dimensione che non si può disegnare. Così analogamente sul video una linea è un insieme di pixel e quindi di punti ciascuno con la sua individualità;

una linea tracciata con il plotter è un insieme infinito di punti. Tanto più sono numerosi i punti sul video tanto più tale ambiente si avvicina a quello geometrico ideale.




Figura 4 - Programma Quattrospirali - Output. Lo Screen 1 del PC IBM consente l'uso di 4 colori contemporaneamente su una pagina video di 320 per 200 pixel; le foto sono state riprodotte in bianco e nero per esigenze tipografiche.

```

100 REM inizializzazione generale
110 SCREEN 1:COLOR 1,0:CLS:LINE (0,0)-(319,199),,B
120 LINE (160,0)-(160,199):LINE (0,100)-(319,100)
200 REM primo quadrante
210 XC=80:YC=50:R=8:SR=.6:P=6.28:SA=P/50
220 FOR A=0 TO 2*P STEP SA:B=A+SA
230 A1=A-INT(A/P)*P:B1=B-INT(B/P)*P
240 R=R+SR:CIRCLE (XC,YC),R,2,A1,B1,.7:NEXT A
300 REM secondo quadrante
310 XC=240:YC=50:R=5:SR=.4:P=6.28:SA=P/50
320 FOR A=3*P TO 0 STEP -SA:R=R+SR
330 LINE (XC,YC)-(XC+R*COS(A),YC-R*SIN(A)*.7),1:NEXT A
400 REM terzo quadrante
410 XC=80:YC=149:R=0:SR=.5:P=6.28:SA=P/50
420 PSET (XC,YC):FOR A=0 TO 3*P STEP SA:R=R+SR
430 LINE-(XC+R*COS(A),YC-R*SIN(A)*.7),3:NEXT A
500 REM quarto quadrante
510 P=3.14:SP=P/10:R=0
520 A1=-2*P+SP:A2=2*P-SP:XC=240:YC=150:SR=2:F=.6
530 FOR A=A1 TO 0 STEP SP:R=R+SR
540 XC=XC+SR*COS(A+P):YC=YC-SR*SIN(A+P)*F
550 X=XC+R*COS(A):Y=YC-R*SIN(A)*F
560 LINE (XC,YC)-(X,Y),2
570 A3=ABS(A+2*P):A4=A3+SP
580 CIRCLE (XC,YC),R,1,A3,A4,F:NEXT A
590 FOR A=0 TO A2 STEP SP:R=R+SR
600 XC=XC+SR*COS(A+P):YC=YC-SR*SIN(A+P)*F
610 X=XC+R*COS(A):Y=YC-R*SIN(A)*F
620 LINE (XC,YC)-(X,Y),2
630 A3=ABS(A+2*P):A4=A3+SP
640 CIRCLE (XC,YC),R,1,A,A+SP,F:NEXT A
990 I$=INKEY$:IF I$="" THEN 990:ELSE CLS:SCREEN 0,0,0:WIDTH 80

```

Figura 5 - Programma Quattrospirali - Listato. La logica è la stessa del programma precedente. Viene usata la istruzione Circle del PC IBM per simulare la spirale.

## Punti linee e superfici

Gli strumenti fondamentali per eseguire qualsiasi disegno sono i punti e le linee (o meglio solo i punti se si considerano le linee come successioni di punti). Questo vale sempre sia per il bambino che disegna su una lavagna con i gessetti colorati, sia per l'architetto che progetta un edificio, sia per i tecnici del Laboratorio di Los Alamos che eseguono la famosa sigla di QUARK.

Avendo la possibilità di tracciare punti singolarmente, facendolo con una certa logica si possono tracciare linee, curve o rette che siano, si possono riempire aree, con tanti punti per eseguire campiture.

I linguaggi grafici hanno sempre una istruzione che permette di tracciare un punto e di cancellarlo.

Tutte le altre funzioni, più potenti, comprendono la prima in quanto qualsiasi funzione grafica opera a livello di pixel.

Così la istruzione LINE (X1, Y1) - (X2, Y2), C,B (nel linguaggio BASICA del PC IBM), che permette di disegnare un rettangolo non fa che individuare tutti i pixel che ne costituiscono i suoi quattro lati.

Se si pone  $X1 = X2, Y1 = Y2$  l'istruzione disegna un singolo pixel. Se si pongono i parametri uguali ai limiti della definizione video si traccia una cornice. Se si passa il parametro F l'istruzione può servire per colorare lo sfondo del colore C.

Quindi in generale dato un problema i metodi per risolverlo sono in genere molti e dipendono dal set di istruzioni grafiche disponibili.

Un'ultima considerazione fatta per i neofiti della Computer Grafica. Vanno sempre separati i metodi risolutivi di un problema dagli strumenti con i quali realizzare la soluzione una volta che sia stata trovata.

Come disegnare un cerchio, per esempio, è un problema risolto fin dai tempi di Euclide ed oltre. Chi con un computer vuole disegnare un cerchio deve innanzitutto sapere come si fa (eventualmente leggendo un libro di geometria) e deve sapere di quali strumenti dispone.

Se dispone di una istruzione di tracciamento segmenti deve spezzettare la circonferenza in tante fette, in cui l'arco sia sostituito da un segmento. L'insieme dei segmenti genera la circonferenza.

Se esiste l'istruzione CIRCLE X, Y, R vuol dire che è il computer stesso che conosce l'algoritmo di individuazione di una circonferenza dato centro e raggio.

## I programmi

Per passare alla parte pratica dell'articolo

lo, ragioneremo un po' con la spirale di Archimede, che è la spirale più elementare e vedremo come la si può tracciare con un computer.

Vedremo tre modi di risolvere il problema che differiscono l'uno dall'altro a seconda degli strumenti software che si utilizzano.




Figura 6 - Programma Plotter - Output. L'output è realizzato utilizzando solo un quarto del formato A3 su cui lavora il Graphtec MP 1000.

```

10 REM INIZIALIZZ
12 D# = CHR# (4) : L = 800 : H = 600 : D = 50 : E = 2 * D
14 X = 0 : Y = 0 : GOSUB 44 : X = 3 * D + 2 * L : Y = 3 * D + 2 * H : GOSUB 52
16 REM PRIMO QUADRANTE
18 P = 2 : GOSUB 56 : X = D : Y = D : GOSUB 44 : X = L : Y = H : GOSUB 52
20 X = 450 : Y = 350 : R = 10 : S = 300 : A = 0 : B = 12000 : C = 0 : GOSUB 60
22 REM SECONDO QUADRANTE
24 P = 3 : GOSUB 56 : X = D : Y = E + H : GOSUB 44 : X = L : Y = H : GOSUB 52
26 X = 450 : Y = 1000 : R = 10 : S = 300 : A = 12000 : B = 0 : C = 300 : GOSUB 60
28 REM TERZO QUADRANTE
30 P = 4 : GOSUB 56 : X = E + L : Y = D : GOSUB 44 : X = L : Y = H : GOSUB 52
32 X = 1300 : Y = 350 : R = 300 : S = R : A = 0 : B = 3600 : C = - 12 : GOSUB 60
34 P = 3 : GOSUB 56 : R = 280 : S = R : A = 150 : B = 3750 : C = - 12 : GOSUB 60
36 REM QUARTO QUADRANTE
38 P = 5 : GOSUB 56 : X = E + L : Y = E + H : GOSUB 44 : X = L : Y = H : GOSUB 52
40 X = 1300 : Y = 1000 : C = 0 : FOR R = 100 TO 300 STEP 30
42 S = R : A = 10 * R : B = 2700 + A : GOSUB 60 : NEXT R : END
44 REM MOVE
46 PRINT D#"PR#4" : PRINT "M" : X : ", " : Y : PRINT D#"PR#0" : RETURN
48 REM DRAW
50 PRINT D#"PR#4" : PRINT "D" : X : ", " : Y : PRINT D#"PR#0" : RETURN
52 REM BOX
54 PRINT D#"PR#4" : PRINT "21" : ", " : X : ", " : Y : PRINT D#"PR#0" : RETURN
56 REM CAMBIO PENNA
58 PRINT D#"PR#4" : PRINT "J" : P : PRINT D#"PR#0" : RETURN
60 REM ARCO/CERCHIO/SPIRALE/POLIGONO
62 PRINT D#"PR#4" : PRINT "W" : X : ", " : Y : ", " : R : ", " : S : ", " : A : ", " : B : ", " : C
64 PRINT D#"PR#0" : RETURN

```

Figura 7 - Programma Plotter - Listato. Da riga 44 in poi ci sono le varie routine riconosciute dal Plotter, nel nostro caso collegato allo Slot 4 dell'Apple II.

1) spirali realizzate con una istruzione che traccia segmenti (utilizzeremo la HPlot dell'Applesoft),

2) spirali realizzate con una istruzione che traccia archi (la CIRCLE del BASIC del PC IBM),

3) spirali realizzate con una istruzione che traccia ... spirali (la "W" del software di base del Plotter Graphtec MP 1000).

Supponiamo di tracciare su un foglio di carta una circonferenza e supponiamo di allargare via via la apertura del compasso, impedendo alla linea che stiamo tracciando di chiudersi.

In questo modo abbiamo realizzato una spirale.

Esistono vari tipi di spirali differenziati dalla legge matematica di variazione del raggio con l'angolo. Es:

spirale di Archimede	$R = A * T$
spirale iperbolica	$R = A / T$
spirale superiore	$R * T^n = A$
spirale logaritmica	$R = K * A^T$

dove in ciascuna equazione, espressa in coordinate polari, viene dato il raggio in funzione dell'angolo T.

Esistono poi spirali nello spazio in cui la legge di variazione riguarda anche la terza coordinata.

Esempio è la spirale cilindrica, così detta perché si svolge su di una superficie cilindrica. È in pratica una circonferenza che non si chiude mai in quanto nel tracciarla ci si muove nello spazio.

Il primo programma si chiama "Dieci Spirali" e disegna nove spirali differenti nel video HGR2 dell'Apple II (listato in fig. 3 e due output dei nove in fig. 1 e fig. 2).

Alcuni dati sono fissi (routine 10-20), c'è poi una routine richiamata ad ogni passaggio da una spirale ad un'altra che serve per pulire il video, fare la cornice, ecc.

Le 9 routine sono a 100, 200, ecc ... Ciascuna mostra una variazione sul tema "Spirale". Le prime si basano su quella di Archimede e cioè il raggio varia linearmente con l'angolo.

Le variazioni consistono nel disegnare i raggi, nello spostare il centro anch'esso con una legge lineare rispetto ad A.

Ci sono poi una spirale iperbolica (riga 700), una spirale superiore (riga 800), e una spirale logaritmica (riga 900).

Il secondo programma "Quattro Spirali" (listato in fig. 5 e output in fig. 4) realizza quattro spirali sullo SCREEN 1 del PC IBM (quello con 320 per 200 pixel di definizione e 4 colori).

La differenza con il programma precedente è che con l'Apple II abbiamo usato

per approssimare la spirale segmenti tracciati con l'istruzione H PLOT. Con l'IBM l'approssimazione è realizzata con archi di cerchio, ma poiché archi successivi sono congiunti l'uno all'altro, sarà il centro che via via si sposta. Lo spostamento sarà verso l'interno, lungo il raggio comune a due archi di cerchio successivi.

La sintassi dell'istruzione CIRCLE del BASIC del PC è:

CIRCLE X,Y,R,A1,A2,F

dove X,Y sono le coordinate del centro, R è il Raggio, A1 e A2 angolo di partenza e di arrivo (con valori accettati da  $-2 * \text{PI}$  a  $2 * \text{PI}$ , cosa che crea un po' di problemi in fase di programmazione). F è un fattore di "schiacciamento" che serve per scalare la Y rispetto alla X. In pratica con  $F = .7$  il formato della spirale diventa proporzionale al formato della cornice.




Figura 8 - Programma Spirale cilindrica - Output. La foto rende perfettamente l'idea del cilindro su cui si svolge la spirale.

```

10 REM inizializzazione costanti
20 KEY OFF:SCREEN 1:COLOR 1,0:CLS:WINDOW (-14,-10)-(14,10):RH=40:D=80
30 TH=1:S1=SIN(TH):C1=COS(TH):PH=1.1:S2=SIN(PH):C2=COS(PH)
40 REM assi cartesiani e cornice
50 LINE (-14,-10)-(14,10),2,B:FOR K=1 TO 3
60 READ X,Y,Z:GOSUB 190:X1=SX:Y1=SY:READ X,Y,Z:GOSUB 190:X2=SX:Y2=SY
70 GOSUB 160:NEXT K
80 REM spirale cilindrica
90 R=2:P=3.1416:AI=-P*5:AF=P*5:SA=P/20
100 X=(AI-P)/3:Y=R*COS(AI):Z=R*SIN(AI)
110 GOSUB 190:X1=SX:Y1=SY:GOSUB 180
120 FOR A=AI TO AF STEP SA
130 X=(A-P)/3:Y=R*COS(A):Z=R*SIN(A)
140 GOSUB 190:X1=SX:Y1=SY:GOSUB 170:NEXT A
150 I$=INKEY$:IF I$="" THEN 150:ELSE END
160 LINE (X1,Y1)-(X2,Y2):RETURN:REM line
170 LINE -(X1,Y1),1:RETURN:REM line
180 PSET (X1,Y1),1:RETURN:REM pset
190 REM da tridi a bidi
200 XE=-X*S1+Y*C1:YE=-X*C1*Y2+Z*S2:ZE=-X*S2*C1-Y*S2*S1-Z*C2+RH
210 SX=D*XE/ZE:SY=D*YE/ZE:RETURN
220 REM data riferimento
230 DATA -7,0,0,7,0,0,0,-10,0,0,10,0,0,0,-5,0,0,5

```

Figura 9 - Programma Spirale cilindrica - Listato. I dati si riferiscono a 6 punti utilizzati per individuare gli assi cartesiani. La routine per passaggio da tridimensionale a bidimensionale è quella classica.

Il terzo programma (listato in fig. 7 e output su carta in fig. 6) utilizza in uscita il plotter Graphtec MP 1000, provato alcuni mesi fa su MCmicrocomputer.

Il software di base di tale macchina riconosce una istruzione con il codice "W" che a seconda dei parametri che gli vengono passati può disegnare archi, cerchi, poligoni, spirali, spezzate, ecc.

Da riga 44 in poi ci sono le routine riconosciute dal plotter e che sono realizzate in Applesoft. In pratica il plotter viene riconosciuto dal computer come se fosse una qualsiasi stampante. Le routine sono:

MOVE spostamento fino ad una certa coordinata senza tracciare,

DRAW come sopra però eseguendo il tracciamento,

BOX disegno di un rettangolo,

PEN per il cambio delle penne (ce ne sono 6)

CIRCLE è la istruzione codice "W", cui vanno passati X,Y centro del cerchio, R,S raggio iniziale e finale, A,B angolo iniziale e finale, dati rispettivamente in decimi di millimetro e in decimi di grado (e sono accettati valori superiori all'angolo giro).

L'ultimo parametro C se è pari a zero non interviene, se è minore di zero indica il numero di fette in cui la spirale viene spezzata, se è maggiore di zero indica l'angolo sotteso dalla corda in cui la linea viene spezzata.

Ovviamente il programma non usa tutte le istruzioni del plotter. Le quattro routine principali ciascuna delle quali si riferisce a uno dei quattro settori del disegno, consistono semplicemente nell'assegnare valori opportuni ai parametri riconosciuti dalla routine plotter.

L'ultimo programma disegna una spirale nello spazio, del quale sono tracciati anche gli assi cartesiani (listato in fig. 9 e output in fig. 8). La spirale è cilindrica ed è realizzata utilizzando la formula della circonferenza, che però non si chiude mai perché via via si solleva, in quanto è resa variabile, linearmente con l'angolo, anche la coordinata Z. Il nome cilindrica deriva dal fatto che in pratica si svolge su di un cilindro.

Nel programma ci sono dapprima una inizializzazione (righe 10-30), poi il disegno degli assi cartesiani (40-70) e infine il disegno della spirale (80-150).

Le routine richiamate sono la 190 che esegue la traduzione da coordinate tridimensionali (X,Y,Z) a bidimensionali (SX,SY) secondo le caratteristiche trigonometriche definite in fase di inizializzazione.

**COMPUTER DIDATTICI MPF  
(MICROPROFESSOR)**  
 MPF I/ P basato su 280 8 bit  
 MPF I/65 basato su 6502 8 bit  
 MPF I/88 basato su 8088 16 bit  
 MPF I/68 basato su 68000 32 bit

# I COMPUTER MPF PER POTER SCEGLIERE



MPF I

**HOME/PERSONAL  
COMPUTER**  
**MPF II**  
 CPU: 6502, 1 Mhz/ROM:  
 16 K con interprete basic  
 apple soft. Monitor,  
 Disassembler / **Ram: 64 K**



MPF II

**MPF III**  
 CPU: 6502, 1 Mhz /  
 ROM: 24 K (con interprete  
 basic) / **RAM: 64 K**  
 dinamiche più 2 K statiche  
 per le 80 colonne di testo /  
**TESTO: 40 x 24 - 80 x 24;**  
**TASTIERA: distaccata con**  
 90 tasti multifunzione.



MPF III

**PERSONAL/PROFESSIONAL COMPUTER**  
**MPF PC - MPF PC/XT**  
 CPU 8088 più 8087 (opzionale)  
 ROM 8 K espandibili a 48 K su scheda  
 RAM 128 K espandibili a 256 K su  
 scheda e 640 K esternamente  
**Disk drive MPF PC 2 disk drive**  
 per 720 K  
**Disk drive MPF PC/XT 1 disk**  
 drive per 360 K più HD da 10  
 Mb.  
**Interfacce resistenti 1**  
 RS232 + centronic +  
 controller disk drive +  
 scheda colore + 4 slot  
 compatibili disponibili  
 utente.  
**Sistema operativo**  
 standard: Concurrent  
 CP/M86 con PC MODE  
**Sistema operativo**  
 opzionale: MS-DOS



MPF PC



MPF PC/XT

**DIGITEK COMPUTER**

VIA VALLI, 28 - 42011 BAGNOLO IN PIANO (Reggio Emilia)  
 Tel. (0522) 61623 r.a. - Telex 530156